
LET FUNCTION

PURPOSE
Defines a function.

DESCRIPTION
DATAPLOT allows users to define their own functions. Functions are defined with a Fortran like syntax. User defined functions can be
used in subsequent PLOT and LET commands (basically any place that a built-in DATAPLOT function can be used).

SYNTAX
LET FUNCTION   <f1>   =    <f2>
where <f1> is the name of the function;
and  <f2> is a functional expression.

EXAMPLES
LET FUNCTION F1 = C*EXP(0.5*((X-U)/SIGM
LET FUNCTION F4 = F1*F2*LOG(F3)
LET FUNCTION G = DERIVATIVE F1 WRT X

NOTE 1
DATAPLOT supports a large number of built-in functions. These are documented in Volume II of the DATAPLOT Reference Manual.
All of DATAPLOT’s built-in functions can be used in user defined functions.

NOTE 2
Functions can be nested. For example:

LET FUNCTION F1 = -0.5*((X**2)+(Y**2))
LET FUNCTION F2 = (1/(2*PI))*EXP(F1)

It is often easier to write a complex function as a series of simpler functions.

NOTE 3
Functions can depend on more than one variable. For example,

LET FUNCTION F1 = -0.5*((X**2)+(Y**2))

This function depends on both X and Y. In practice, user defined functions with one or two variables are common while more than two
variables are fairly rare.

NOTE 4
To evaluate a function at a sequence of points, do something like the following:

LET FUNCTION F1 = 3*X**2 + 2*X + 4
LET X = SEQUENCE -3 .1 3
LET Y1 = F1

This evaluates function F1 at the points defined in the variable X. The computed values are stored in the variable Y1 (it has the same
length as X).

There is a distinction for functions with more than one variable. For example, the following only evaluates the function at the points     (-
2,-2), (-1.9,-1.9), ... , (2,2):

LET FUNCTION F1 = -0.5*((X**2)+(Y**2))
LET FUNCTION F2 = (1/(2*PI))*EXP(F1)
LET X = SEQUENCE -2 .1 2
LET Y = SEQUENCE -2 .1 2
LET Z = F2

If you really want a grid from -2 to 2 in both the X and Y directions, do the following:

LET FUNCTION F1 = -0.5*((X**2)+(Y**2))
LET FUNCTION F2 = (1/(2*PI))*EXP(F1)
LET TEMP = SEQUENCE -2 .1 2
LET NPTS = SIZE TEMP

LET NPTS2 = NPTS*NPTS
LET X = SEQUENCE -2 .1 2 FOR I = 1 1 NPTS2
LET Y = SEQUENCE -2 NPTS .1 2
LET Z = F2

NOTE 5
The total number of characters that DATAPLOT can use for storing functions and strings is set when DATAPLOT is built. The current
default is 10,000 characters. Previous versions set this limit at 1,000 characters. This limit applies to the combined number of characters
for all functions and strings.

NOTE 6
Error checking (e.g., matching parenthesis, valid number of arguments for a built-in function) is performed when the function is
evaluated, not when it is created.

DEFAULT
None

SYNONYMS
None

RELATED COMMANDS
LET = Carries out variety of operations on variable, parameters, and functions.
STATUS = Displays dimension, variables, parameters, functions, etc.
WRITE = Writes variables, parameters, functions to either the screen or to file.

APPLICATIONS
User defined functions

IMPLEMENTATION DATE
Pre-1987

PROGRAM
LET FUNCTION F1 = SIN(X)*COS(X)
LET START = -PI/2
LET STOP = PI/2
LET X = SEQUENCE START 0.1 STOP
LET D1 = DERIVATIVE F1 WRT X
LINE SOLID DASH
LET Y1 = F1
TITLE PLOT OF FUNCTION AND ITS DERIVATIVE
PLOT Y1 D1 VS X

